

第5章 モナドと命令型言語の意味

IO は効率のために Haskell の処理系で特別扱いされる組込みのモナドであるが、他の言語の副作用を模倣するためにユーザーがモナドを定義することも可能である。モナドを用いる利点は、“計算”の意味が変わっても、モナドの標準的な関数 *return* と ($\gg=$) のみを用いている部分は、変更する必要がないところである。

以下では、簡単な命令型プログラミング言語（つまり副作用を持つ言語）を定義し、モナドを利用してその意味を Haskell で与えることにする。Haskell で意味を与えるということは、等式による推論が可能になるということである。具体的には命令型プログラミング言語から Haskell へのコンパイラーを作成する。

5.1 Util コンパイラー

この節では、コンパイラーの実装を紹介していく。実装の詳細は把握しなくても、ソースプログラムとターゲットプログラムを比べれば、副作用を持つプログラミング言語がどのように変換されるか感覚的に掴めるはずである。

簡単な言語からはじめて様々な特徴をもつ言語を定義していく。名前がないと不便なので、これらの命令型言語を Util (Util: Tiny Imperative Language)¹ と呼び、必要により、UtilErr, UtilST, UtilCont, ... などのようにバージョンを表す接尾語をつけることにする。いろいろな特徴を導入していくにつれ、その“計算”を表すモナドの定義が変わることになる。

実際のコンパイラーにはフロントエンド、つまり _____ (空欄 5.1.1) や _____ (空欄 5.1.2) が必要である。字句解析や構文解析の原理は Haskell でも C 言語などの命令型言語で記述するときと変わりはない。再帰下降構文解析法（あるいは LR 構文解析法）などの方法を利用する。(ただし、再帰下降法で構文解析部を記述するときに、後述のようにモナドを利用することができる。)

しかし、ここではこれらフロントエンドの作り方は既知のものとして、構文木ができた状態から話をはじめることにする。

¹いわゆる再帰的頭字語 (recursive acronym) である。PHP, GNU などの略語の由来も参照すること。

5.1.1 構文規則

Util の構文木のデータ構造として、次のような Haskell データ型を使用する。
ファイル RecType.hs

```
1  type Decl = (String, Expr)
2  data Expr = Const Target          -- 定数 (Target は後述)
3             | Var String           -- 変数
4             | If Expr Expr Expr    -- if 文
5             | While Expr Expr      -- while 文
6             | Begin [Expr]         -- ブロック
7             | Let [Decl] Expr      -- let 式 (関数定義)
8             | Val Decl Expr        -- val 式 (変数定義)
9             | Lambda String Expr   -- ラムダ式
10            | Delay Expr           -- delay 式 (後述)
11            | App Expr Expr        -- 関数適用
12            deriving Show
```

つまり、式 (Expr) とは、定数 (Const) または、変数 (Var) または、if 式 (If)、let 式 (Let)、ラムダ式 (Lambda)、関数適用 (App) などからなる。(あとから必要に応じて構文要素を追加することにする。)

Util の具体的な構文としては次のような BNF で定義されていると仮定する。(演算子の優先順位なども適切に宣言されているとする。)

```
Expr  → Const | Var | ( Expr )
      | if Expr then Expr else Expr | while Expr do Expr
      | begin Exprs end
      | let Decls in Expr | val Binds in Expr
      | \ Vars -> Expr | Expr Expr
      | Expr + Expr | Expr * Expr | ... (他の中置演算子) ...

Exprs → Expr | Expr ; Exprs
Decl  → Vars = Expr
Decls → Decl | Decl ; Decls
Bind  → Var <- Expr
Binds → Bind | Bind ; Binds
Vars  → Var | Var Vars
```

ここに示されていないが、定数 *Const* と変数 *Var* の字句の定義は Haskell と同じとする。ただし、`_` (アンダーバー) から始まる変数名はコンパイラ内部で使用するために予約済みとする。Haskell と異なり、`while ~ do` 式や `begin ~ end` 式があるところが命令型言語らしいところである。

5.1.2 字句解析・構文解析関数

次のような関数が既に定義されているものと仮定する。

```
myParse :: String -> Expr  -- 字句解析・構文解析の関数
```

- “`val x <- 2 * 2 in val y <- x * x in y * y`” というソースプログラムは `Expr` 型のデータとして次のように構文解析される。

```
Val ("x", (App (App times (Const (TLit (Int 2))))
              (Const (TLit (Int 2)))))
      (Val ("y", (App (App times (Var "x")) (Var "x")))
          (App (App times (Var "y")) (Var "y"))))
```

ただし、`times` は `*` に対応する `Expr` の式である。

- “`\ f -> \ x -> f x`” という式は、

```
Lambda "f" (Lambda "x" (App (Var "f") (Var "x")))
```

というデータに構文解析される。

- `&&`, `||` は、それぞれ、

```
b1 && b2 ⇨ if b1 then b2 else False
```

```
b1 || b2 ⇨ if b1 then True else b2
```

という糖衣構文であるように構文解析されるようにしておく。

5.1.3 ターゲット言語

コンパイラーのターゲット言語である Haskell のサブセットの構文木を表現する型 `Target` 型を定義しておく。

ファイル `Target.hs`

```
1  type TDecl = (String, Target)
2  data Target = TLit Literal           -- 定数
3              | TVar String           -- 変数
4              | TIf Target Target Target -- if文
5              | TLet [TDecl] Target   -- let式(変数・関数定義)
6              | TLambda1 String Target -- ラムダ式
7              | TApp1 Target Target   -- 関数適用
8              | TReturn Target        -- return に相当
9              | TBind Target Target   -- (>>=) に相当
10         deriving (Show,Eq)
11  data Literal = Str String | Int Integer | Frac Rational
12              | Char Char           deriving (Show,Eq)
```

Util のコンパイラーとは次のような型を持つ関数である。

```
comp :: Expr -> Target  -- コンパイラー
```

5.1.4 抽象構文と具象構文

ところで、上の Util の構文規則は _____ (空欄 5.1.3) (ambiguous) である。つまり `1 + 2 * 3` は足し算と掛け算のどちらを先に実行するか、決定できない。通常は曖昧さを避けるために、

関数 `comp` を理解するために、変換前と変換後を代数的データ型ではなく、それぞれ Util と Haskell の文法で記述したのが次の表である。(詳細はソースプログラムを参照すること。)

この表のなかでソース中で *Italic* フォントで示されている m, n などは任意の Util の式で、ターゲット中で \dot{m}, \dot{n} のように ` (ドット) が付いている式は、その `comp` による変換後の Haskell の式を表す。なお、`delay` については内部的に使用されるので、実際には Util のソースプログラムに現れることはない。

ソース (Util)	ターゲット (Haskell)
<code>c</code> (ただし c は定数)	<code>return c</code>
<code>x</code> (ただし x は変数)	<code>return x</code>
<code>val x <- m in n</code>	<code>\dot{m} >>= \ x -> \dot{n}</code>
<code>let f = \ x -> m g = \ y -> n in k</code>	<code>let f = \ x -> \dot{m} x = \ y -> \dot{n} in \dot{k}</code>
<code>fa</code>	<code>\dot{f} >>= \ _g -> \dot{a} >>= \ _x -> _g _x</code>
<code>\ x -> m</code>	<code>return (\ x -> \dot{m})</code>
<code>delay m</code>	<code>return \dot{m}</code>
<code>if c then t else e</code>	<code>\dot{c} >>= \ _b -> if _b then \dot{t} else \dot{e}</code>
<code>while c do t</code>	<code>let _while = \dot{c} >>= \ _b -> if _b then \dot{t} >>= \ _ -> _while else return () in _while</code>
<code>begin s; t; u end</code>	<code>\dot{s} >>= \ _ -> \dot{t} >>= \ _ -> \dot{u}</code>

要点は、変数や定数の出現など“副作用”が発生しないところには `return` が付くこと、関数適用は (`>>=`) を使った式に翻訳されること、などである。

また、Util プログラム中の `+`, `-`, `*` などの二項演算子は、他の関数が `return (\ x -> ...)` という形に変換されること、戻り値はアクションを持たないことから、同名の Haskell 内のオペレーターを用いて、それぞれ

```
return (\ x -> return (\ y -> return (x + y)))
return (\ x -> return (\ y -> return (x - y)))
return (\ x -> return (\ y -> return (x * y)))
```

という Haskell の式に置換するようにしておく。すると、`comp` 関数による他の部分の変換と整合する。

ソース (Util)	ターゲット (Haskell)
<code>⊗</code> (ただし <code>⊗</code> は二項演算子)	<code>return (\ x -> return (\ y -> return (x ⊗ y)))</code>

この `comp` を用いて、例えば次の Util プログラムを変換²すると

```
fact = \ n -> if n == 0 then 1 else n * fact (n - 1)
```

次のような Haskell のプログラムが得られる。

```
1 fact = \ n ->
2   ((return (\ x -> return (\ y -> return (x == y))) >>=
3     \ _f -> return n >>= \ _x -> _f _x)
4     >>= \ _f -> return 0 >>= \ _x -> _f _x)
5     >>=
6     \ _b ->
7       if _b then return 1 else
8         (return (\ x -> return (\ y -> return (x * y))) >>=
9           \ _f -> return n >>= \ _x -> _f _x)
10          >>=
11          \ _f ->
12            (return fact >>=
13              \ _f ->
14                ((return (\ x -> return (\ y -> return (x - y)))
15                  >>= \ _f -> return n
16                    >>= \ _x -> _f _x)
17                  >>= \ _f -> return 1
18                    >>= \ _x -> _f _x)
19                  >>= \ _x -> _f _x)
20                >>= \ _x -> _f _x)
```

これは多くの冗長な部分を含んでいるので、前述の monad law などを利用して単純化すると、多くの `return` や (`>>=`) が消えて、次のような Haskell の式が得られる。

```
1 fact = \ n -> if n == 0 then return 1 else
2           fact (n - 1) >>= \ _x ->
3           return (n * _x)
```

5.2 最初のバージョン - Util1

最初のバージョン Util1 では、モナドはトリビアルな計算（何もしない計算）としておく。つまり、Util1 は副作用を持たない言語である。

ファイル Id.hs

```
1 newtype I a = I a
2
3 instance Monad I where
4   return a = I a
5   (I m) >>= k = k m
```

ここで、 (空欄 5.2.1) は、Haskell の新しい型の宣言の形式の一つである。data 宣言と似ているが、フィールドが一つの構成子を一つしか持つことがで

²ただし、この `fact` 関数は副作用を含んでいないので、この変換自体にはあまり意味はない。

きない。また `data` 宣言の構成子と異なり、`newtype` 宣言で導入される構成子は型変換の意味しか持たず、実行時の計算を伴わない。また、型の別名を宣言する `type` 宣言との違いは、型の変換が構成子によって明示的になる点である。(型クラスのインスタンスには、`type` 宣言で導入された型の別名は指定できない。例えばリストに通常の辞書式順序と異なる順序 (`Ord` のインスタンス宣言) を与えたいときは `newtype` を使って別名を用意し、別名に対してインスタンス宣言する必要がある)

上記の `newtype` 宣言では型構成子と構成子に同じ `I` という名前を使っているが、文脈でどちらか判断することができるので問題ない。

なお、ある型構成子を `Monad` のインスタンスと宣言するとき、同時に `Monad` のスーパークラスである `Functor` と `Applicative` に対してもインスタンス宣言をする必要がある。今後紹介するモナドに対してもすべて同一なので `I` に対するインスタンス宣言だけを代表として紹介しておく。

ファイル `Id.hs`

```
1 instance Functor I where
2   fmap f m = m >>= \ x -> return (f x)
3
4 instance Applicative I where
5   pure = return
6   g <*> m = g >>= \ f -> m >>= \ x -> return (f x)
```

このとき、

```
fact n = if n == 0 then 1 else n * fact (n - 1)
```

という `Util` プログラムをコンパイルして実行する (`unI (fact 9)`) と、同じプログラムを `Haskell` として実行したときと全く同じ `362880` という値になる。

ただし、`unI` は次のように定義された型変換関数である。

ファイル `Id.hs`

```
1 unI :: I a -> a
2 unI (I a) = a
```

5.3 UtilST – 状態の導入

`Util` に更新 (代入) 可能な状態の概念を導入する。ここで紹介する例では、2つだけ更新可能な“参照” `xP` と `yP` を導入することにする。2 という数は別に本質的なものではなく、いくつにすることも可能である。参照は `set` で値を代入し、`get` で値を取り出すことができる。

例えば、

```
begin set xP 1; set xP (get xP+3); get xP end
```

という `UtilST` プログラムを評価すると、 (空欄 5.3.1) という結果が得られる。

状態を導入するために、やはり“計算の型”を定義する必要がある。まず次の ST を定義する。

ファイル ST.hs

```
1  newtype ST s a = ST (s -> (a,s))
2
3  unST :: ST s a -> s -> (a,s)
4  unST (ST s) = s
5
6  instance Monad (ST s) where
7    return a = ST ( \ s0 -> (a,s0) )
8    (ST m) >>= k = ST ( \ s0 -> let { (a,s1) = m s0 }
9                               in unST (k a) s1 )
10   -- ST, unST がなければ、次のようになる
11   -- m >>= k = \ s0 -> let { (a,s1) = m s0 } in k a s1
```

このモナドは State Transformer モナドと呼ばれる。return a は状態 (s) の変更を行わず、a をそのまま返す計算である。このモナドの m >>= k は、m で変更された状態 (s1) をそのまま、k に受渡す計算である。

問 5.3.1 ST に対して、次の monad law が成り立つことを確認せよ。

$$\begin{aligned}(\text{return } a) >>= k &= k a \\ m >>= (\backslash a \rightarrow \text{return } a) &= m \\ (m1 >>= k1) >> k2 &= m1 >>= (\backslash a \rightarrow (k1 a >>= k2))\end{aligned}$$

参照は次のように定義する。

ファイル MyState.hs

```
1  type Pos s a = s -> (a, a -> s)
2
3  xP :: Pos (x,y) x
4  xP = \ (x,y) -> (x, \ x1 -> (x1,y))
5
6  yP :: Pos (x,y) y
7  yP = \ (x,y) -> (y, \ y1 -> (x,y1))
```

ここで xP (yP) はペアの第 1 (第 2) 成分にアクセスするための参照である。参照に対する読み出し・書き込みのメソッドは、後で別のモナドでも使用するの、型クラス MyState のメソッドとして定義しておくことにする。

ファイル MyState.hs

```
1  class MyState m where
2    get :: Pos s a -> m s a
3    set :: Pos s a -> a -> m s ()
```

そして ST をこの MyState クラスのインスタンスとして宣言する。

ファイル ST.hs

```

1 instance MyState ST where
2   get p = ST (\ s -> (fst (p s), s))
3   set p v = ST (\ s -> ((), snd (p s) v))
4
5   -- 例えば get xP ≡ ST (\ (x,y) -> (x,(x,y)))
6   -- 例えば set xP x1 ≡ ST (\ (x,y) -> ((),(x1,y)))

```

ここで set は状態を書き換える、また get は状態の値の一部を複製する関数である。

UtilST の set, get, ... という関数は、Haskell の set, get, ... にコンパイルされるようにしておく

ソース (Util)	ターゲット (Haskell)
set p m	p >>= \ _p -> m >>= \ _x -> set _p _x
get p	p >>= \ _p -> get _p

UtilST プログラム (右は対応する C プログラム):

<pre> 1 fact n = begin 2 set xP 1; set yP n; 3 while get yP > 0 do begin 4 set xP (get xP * get yP); 5 set yP (get yP - 1) 6 end; 7 get xP 8 end </pre>	<pre> 1 int fact(int y) { 2 int x = 1; 3 while (y > 0) { 4 x = x * y; 5 y = y - 1; 6 } 7 return x; 8 } </pre>
--	--

をコンパイルすると次のような Haskell の関数 (一部、見易くするために変数名の変更などを行っている) になる。

```

1 fact n = set xP 1 >>= \ _ ->
2   set yP n >>= \ _ ->
3   (let _while = get yP >>= \ y ->
4     if y > 0 then
5       get xP >>= \ x ->
6       get yP >>= \ y ->
7       set xP (x * y) >>= \ _ ->
8       get yP >>= \ y ->
9       set yP (y - 1) >>= \ _ ->
10      _while
11     else return ()
12   in _while) >>= \ _ ->
13   get xP

```

ここで

```

1 evalST st s = fst (unST st s)

```

と定義すると、fact 9 を実行する (evalST (fact 9) (0,0)) と、その結果は 362880 になる。

階乗の場合、普通に関数的な定義のほうが簡潔だが、パラメーターの数が多い場合などは、このような命令的な書き方が簡潔になる場合もありうる。

この ST の定義では、エラー処理を考慮していない。エラー処理を行なうためには、この ST と (後述する) Maybe の定義を合成する必要がある。参考までに、次のようなモナドになる。

ファイル EST.hs

```
1  newtype EST s a = EST (s -> Maybe (a,s))
2
3  unEST :: EST s a -> s -> Maybe (a,s)
4  unEST (EST m) = m
5
6  instance Monad (EST s) where
7    return a = EST (\ s -> return (a,s))
8    (EST m) >>= k = EST (\ s0 -> case m s0 of
9                                Just (a,s1) -> unEST (k a) s1
10                               Nothing    -> Nothing)
```

問 5.3.2 次の C の関数とほぼ同等な UtilST の関数、または、ST モナドを用いた Haskell の関数を定義せよ。

```
1.  int foo(int n) {
      int i = 1, j = 1;
      while (i < n) {
          i = i + j;
          j = i - j;
      }
      return i;
  }
```

```
2.  int bar(int n) {
      int i = 0;
      while (n > 1) {
          i = i + 1;
          n = n / 2;
      }
      return i;
  }
```

整数の割り算は Haskell では 'div' 演算子 になる。

5.4 (参考) UtilIO – 入出力の導入

入出力は、入出力ストリームを状態の一種と考えれば、5.3 節の UtilST と同じ方法で取り扱うことができる。

計算のモナドの定義は 5.3 節と基本的に同じだが、状態に入力と出力のストリームを表す String 型の部分を追加しておく。

ファイル MyStream.hs

```
1 type WithIO s = (s,String,String)
```

ファイル MyIO.hs

```
1 newtype MyIO s a = MyIO (WithIO s -> (a, WithIO s))
2
3 unMyIO :: MyIO s a -> WithIO s -> (a, WithIO s)
4 unMyIO (MyIO m) = m
```

参照のプリミティブの定義は次のようになる。

ファイル MyIO.hs

```
1 instance MyState MyIO
2   get p    = MyIO (\ (s,i,o) -> (fst (p s),(s,i,o)))
3   set p v  = MyIO (\ (s,i,o) -> ((),(snd (p s) v,i,o)))
```

入出力に関するプリミティブも後で別のモナドで使用するのので、型クラス MyStream のメソッドとして定義しておく。

ファイル MyStream.hs

```
1 class MyStream m where
2   readChar  :: m Char
3   eof       :: m Bool
4   writeStr  :: String -> m ()
```

ファイル MyIO.hs

```
1 instance MyStream (MyIO s) where
2   readChar  = MyIO (\ (s,c:cs,o) -> (c,(s,cs,o)))
3   eof       = MyIO (\ (s,i,o) -> (null i,(s,i,o)))
4   writeStr v = MyIO (\ (s,i,o) -> ((),(s,i,o ++ v)))
```

ここで readChar は入力ストリームから 1 文字を取出す。また writeStr str は出力ストリーム o に str を追加する³。そして write という関数を次のように定義しておく。

ファイル MyStream.hs

```
1 write :: (Show v, MyStream m) => v -> m ()
2 write v = writeStr (show v)
```

³ただし++の計算量は左オペランドの長さに比例するので、この定義のように文字列の後ろに新しい文字列を追加(++)していくと、出力文字列が長くなるにしたがって効率が悪くなる。これを避けて効率の良い定義を与えることも可能であるが、ここでは簡単のために++を使った定義を採用する。

すると、次の UtilIO プログラム (ただし、// は整数の除算を表す演算子とする。)

```
1 foo n = begin
2   set xP n;
3   while get xP > 0 do begin
4     write (get xP % 10);
5     set xP (get xP // 10)
6   end
7 end
```

をコンパイルした結果は、

```
1 foo n = set xP n >>= \ _ ->
2         let _while
3           = get xP >>= \ _x ->
4             if _x > 0 then
5               get xP >>= \ _x ->
6                 write (_x 'mod' 10) >>= \ _ ->
7                 get xP >>= \ _x ->
8                 set xP (_x 'div' 10) >>= \ _ ->
9                 _while
10            else return ()
11         in _while
```

となり、補助関数を以下のように定義したとき、

```
evalMyIO e s = let (_,(_,_,o)) = unMyIO e (s,"","") in o
```

関数 foo を evalMyIO (foo 12345) (0,0) のように実行すると、出力は "54321" になる。

問 5.4.1 次の C の関数とほぼ同等な UtilIO の関数、または、MyIO モナドを用いた Haskell の関数を定義せよ。

```
1. int baz(int n) {
   int i, j;
   for (i = 0; i < n; i++) {
     for (j = 0; j <= i; j++) {
       printf("*");
     }
     printf("\n");
   }
   return i;
}
```

```
2. int qux(int n) {
   int i, j;
   for (i = 0; i < n; i++) {
     printf("*");
     if (i % 3 == 0) {
       printf("!");
     }
   }
}
```

```

    printf("\n");
  }
  return i;
}

```

5.5 UtilErr – エラー処理の導入

次に Util にエラー処理を導入する。UtilErr は、生真面目に (?) エラー処理を行ない、部分式にエラーがあれば式全体もエラーになるようにする。この場合、UtilErr は (Haskell のような) 遅延評価ではなくて、関数の引数を必ず先に評価する _____ (空欄 5.5.1) (eager evaluation) を模倣することに注意する必要がある。

エラーと正常な振舞いを区別するために、次のような標準ライブラリに用意されているデータ型 Maybe を使用する。

```

-- Prelude に定義済み
data Maybe a = _____ | _____ deriving (Show,Eq)

```

正常な振舞いは Just という構成子で表す。エラーの場合は Nothing という構成子を用いる。この型に対して次のようなインスタンス宣言がされている。

```

1  -- Prelude に宣言済み
2  instance Monad Maybe where
3    return a = Just a
4    (Just a) >>= k = _____
5    Nothing >>= k = _____

```

このモナドの $m \gg= k$ は、まず m を計算し、その計算が正常終了すれば、その値を k という関数に渡す。しかし、いったん m でエラーが起こると、 k は評価されず、_____ (空欄 5.5.2) していくことを表している。

問 5.5.1 Maybe に対して、次の monad law が成り立つことを確認せよ。

$$\begin{aligned}
 (\text{return } a) \gg= k &= k \ a \\
 m \gg= (\lambda a \rightarrow \text{return } a) &= m \\
 (m1 \gg= k1) \gg k2 &= m1 \gg= (\lambda a \rightarrow (k1 \ a \gg= k2))
 \end{aligned}$$

さらに、MonadPlus というクラスに属するいくつかのメソッドを用意する。ここで mzero は _____ (空欄 5.5.3) 状況をつくり出すときに用いる。

```

1  -- モジュール Control.Monad に定義済み
2  class Monad m => MonadPlus m where
3    mzero :: m a
4    mplus :: m a -> m a -> m a
5
6  -- モジュール Control.Monad に宣言済み
7  instance MonadPlus Maybe where
8    mzero = Nothing
9    ...   -- mplus の定義は後掲

```

“プリミティブ関数”もエラー処理を利用するように書き換えることができる。例えば、割り算 (/) は次のような Haskell の式に置換されるようにする。

```
\ x -> return (\ y -> if y == 0 then mzero
                    else return (x / y))
```

これで0で割ろうとした場合にはエラーが報告される。

例えば

```
(\ x -> 0) (1 / 0)
```

のような式は、Haskell では $1/0$ の部分式は _____ (空欄 5.5.4) に全体の結果が0となるが、UtilErr では次のような Haskell プログラムに翻訳され、

```
1 (if 0 == 0 then mzero
2   else return (1 / 0)) >>= \ _x ->
3   return 0
```

実行すると (エラーが起こったことを表す) _____ (空欄 5.5.5) という結果になる。

5.6 例外処理の導入

例外のモナド Maybe を利用して、Java の try ~ catch のように例外を捕捉する構文を導入することも可能である。

Util の BNF には以下の構文を追加する。

$$Expr \rightarrow \dots \mid \mathbf{try} \ Expr \ \mathbf{catch} \ Expr$$

“**try** *m* **catch** *h*” は *m* を評価し、エラーがなかった場合は、その戻り値を **try** 式の戻り値とする。しかし *m* の評価中にエラーが生じた場合は、*h* を評価する。Util の “**try** *m* **catch** *h*” は “*m* ‘mplus’ *h*” という式として構文解析されるようにしておく。

また、fail という Util の関数は、Haskell の mzero を返す関数にコンパイルされるようにしておく。この関数は、Java の throw 文に対応する。

ソース (Util)	ターゲット (Haskell)
try <i>m</i> catch <i>n</i>	<i>m</i> ‘mplus’ <i>n</i>
fail ()	mzero

Maybe に対する mplus は第1引数を実評価し、_____ (空欄 5.6.1) 第2引数を実評価する関数である。

```
1 instance MonadPlus Maybe where
2   ...      -- mzero は紹介済
3   Just a 'mplus' _ = Just a
4   Nothing 'mplus' m = m
```

例えば

```
bar n = try 1 / n catch 99999
```

という UtilErr プログラムをコンパイルすれば、出力される Haskell プログラムは、

```
1 bar n = (if n == 0 then mzero else return (1 / n))
2         'mplus' (return 99999)
```

であり、bar 0 の結果は Just 99999.0 である。

5.7 UtilNonDet – 非決定性の導入

ここからは対象言語 Util に非決定性を導入する。非決定性 (nondeterminism) とはプログラムの動作に (空欄 5.7.1) が存在することを言う。ある選択肢を選んだ結果、計算が失敗する場合がある、その場合は前の選択肢に戻って計算をやり直す (バックトラック)。非決定性は探索型のパズル・ゲームや構文解析プログラムなどで利用できる。バックトラックをプリミティブな機能として提供する言語としては、論理型言語の (空欄 5.7.2) が有名である。

非決定性の“計算”のモナドは通常のリストを使用する。

```
1 -- Prelude で宣言済み
2 instance Monad [] where
3   return a = [a]
4   [] >>= k = []
5   (x:xs) >>= k = k x ++ (xs >>= k)
6
7 -- モジュール Control.Monad で宣言済み
8 instance MonadPlus [] where
9   mzero = []
10  mplus = (++)
```

このモナドは複数の選択肢を単にリストとして表現している。

問 5.7.1 [] に対して、次の monad law が成り立つことを確認せよ。

$$\begin{aligned} &(\text{return } a) \gg= k = k \ a \\ m \gg= (\lambda a \rightarrow \text{return } a) &= m \\ (m1 \gg= k1) \gg k2 &= m1 \gg= (\lambda a \rightarrow (k1 \ a \gg= k2)) \end{aligned}$$

実はリストに対する return と (>>=) は、リストの内包表記を説明する時に使った `unit :: a -> [a]` と `bind :: [a] -> (a -> [b]) -> [b]` と全く同一の関数である。

計算の失敗は空リストで表される。

UtilNonDet は、構文は UtilErr と同じである。つまり、以下の構文を持つ。

$$Expr \rightarrow \dots \mid \text{try } Expr \text{ catch } Expr$$

UtilNonDetの `try m catch h` は、`m` を評価し、`h` は“バックトラック”が起こったときに評価される。

UtilNonDet プログラム

```
test0 = (try 2 catch 3) * (try 5 catch 7)
```

をコンパイルすると、次の Haskell プログラムが得られる。

```
1 test0 = (return 2 'mplus' return 3) >>= \ x ->
2         (return 5 'mplus' return 7) >>= \ y ->
3         return (x * y)
```

この、`test0` は `[x * y | x <- [2,3], y <- [5,7]` というリスト内包表記と同じ意味になる。そして `test0` は、`[10,14,15,21]` となる。

また、次の UtilNonDet プログラム

```
test1 n = (try 1 catch 2) / (try n catch 4)
```

をコンパイルすると、次の Haskell プログラムが得られる。

```
1 test1 n = (return 1 'mplus' return 2) >>= \ x ->
2         (return n 'mplus' return 4) >>= \ y ->
3         if y == 0 then mzero else return (x / y)
```

そして `test1 2` は、`[0.5,0.25,1.0,0.5]`、`test1 0` は `[0.25,0.5]` となる。失敗している計算については結果に現れていないことに注意する。同じプログラムを UtilErr でコンパイルすると、UtilErr ではバックトラックが起これないので、`test1 0` は全体が失敗 (Nothing) に終わる。

なお、次の `head` を用いてリストの頭部を取ることににより、成功した最初の計算だけを返すことも可能である。

```
1 -- Prelude に定義済み
2 head :: [a] -> a
3 head (x:_) = x
```

このとき `head (test1 0)` の値は `0.25` となる。この場合、Haskell が [\(空欄 5.7.3\)](#) を採用しているため、他の選択肢の計算は行なわれない。そのため選択肢が無数あるような場合でも最初の選択肢の計算結果を出力することができる。

8 クイーンの問題も非決定性を用いて記述することができる。

```
safel xs n m = null xs ||
                val y <- head xs;
                ys <- tail xs in
                y /= n && y /= n + m && y /= n - m
                && safel ys n (m + 1);

safe xs n = safel xs n 1;

range i j = if i > j then fail ()
            else try i catch range (i + 1) j;
```

```
queen n = if n == 0 then []
         else val p <- queen (n - 1);
              n <- range 1 8 in
              if safe p n then (n:p) else fail ()
```

そして queen 8 は、[4,2,7,3,6,8,5,1] から始まる 92 個の解を返す。

問 5.7.2 非決定性と状態の両方の特徴を持つ計算の型として、

```
1 newtype STL s a = STL (s -> ([a],s))
2 newtype LST s a = LST (s -> [(a,s)])
```

の 2 つのバリエーションが考えられる。このそれぞれに対して、コンパイラーの定義を完成させ、2 つの違いを説明せよ。

5.8 Prolog と論理変数

論理型言語の Prolog には非決定性の他に、 (空欄 5.8.1) という特徴がある。論理変数とは、最初は値が定まっておらず、単一化の制約により徐々に値が具体化していく変数である。何度も代入できる命令型言語の変数とも、一度初期化すれば二度と代入ができない関数型言語の変数とも異なる。Haskell では論理変数自体は、命令型言語を模倣するために使った State Transformer モナドと同様の方法で模倣することができる。

例えば、Prolog では、リストを接続 (append) するプログラムは次のように記述される。

```
1 myAppend([H|X], Y, [H|Z]) :- myAppend(X, Y, Z).
2 myAppend([], Y, Y).
```

これは、1 番目の引数と 2 番目の引数を接続した結果が 3 番目の引数になる、という関係を表している。

この myAppend に対して、[1,4,3] と [5,3] の接続を求めるには、次のような問合せ (呼出し) をする。

```
?- myAppend([1,4,3], [2,5], R).

R = [1,4,3,2,5] ;

No
```

さらに Prolog のおもしろいところは myAppend の逆向きの計算もできるということである。

```
?- myAppend(A, B, [1,2,3]).

A = []
B = [1,2,3] ;
```

```

A = [1]
B = [2,3] ;

A = [1,2]
B = [3] ;

A = [1,2,3]
B = [] ;

No

```

この実行例では接続して [1,2,3] になる 2 つのリストの、すべての可能性を求めていることになる。ユーザーが「;」を入力するたびにバックトラッキングが起こり別解を表示する。

MicroKanren.hs⁴ という Haskell のライブラリでは、非決定性と論理変数にプラスアルファの機能を加えたモナドが定義されている。ここで詳細な説明には立ち入らないが、このモナドは State Transformer と非決定性を組み合わせたものである。このライブラリを使えば、上の Prolog の myAppend プログラムに相当するプログラムを Haskell で次のように記述できる。

```

1 myAppend a b ab =
2   do { h <- fresh; t <- fresh; res <- fresh;
3       ht <- cons h t; ht === a;
4       hres <- cons h res; hres === ab;
5       myAppend t b res }
6   'mplus'
7   do { n <- nil; n === a; b === ab }

```

ここで、fresh は新しい論理変数を生成する関数であり、=== は両辺が単一化されることを示す演算子である。

ちなみに、Util の文法では次のように書くことができる。

```

1 myAppend a b ab =
2   try val h = fresh() in val t = fresh() ;
3     res = fresh() in
4     begin
5       cons h t === a; cons h res === ab;
6       myAppend t b res
7     end
8   catch begin nil() === a; b === ab end

```

この myAppend に対して次のようなプログラムを実行する。ただし、toLVList は通常のリストの型から論理変数を含むリストの型への変換、fromLVList はその逆である。ただし fromLVList c xs は xs のなかの未定の論理変数を c に置き換えた通常のリストを返す。

```

1 exampleAppend a =

```

⁴<https://github.com/rntz/ukanren>

```
2   val xs = fresh();
3     ys = fresh();
4     zs = toLVList [1,2,3] in begin
5     myAppend xs ys zs;
6     (fromLVList 0 xs,fromLVList 0 ys)
7   end
```

その結果は $[([], [1,2,3]), ([1], [2,3]), ([1,2], [3]), ([1,2,3], [])]$ になる。

5.9 さらに詳しく知りたい人のために ...

Parsec [1] はモナドを利用した有名なパーサーライブラリーである。[3] にも、モナドを用いてパーサーを構築する技法の解説がある。[2] はモナドを用いてインタプリターを構築する方法を解説している。[4] は、Prolog のカット等のオペレーターの意味を整理している。[5] は、MiniKanrenT で使われているモナドの解説である。

この章の参考文献

- [1] Daan Leijen and Erik Meijer, “Parsec: Direct Style Monadic Parser Combinators for the Real World”
Technical Report UU-CS-2001-35, Dept. of Comp. Sci, Universiteit Utrecht, 2001 年, <http://www.cs.uu.nl/people/daan/parsec.html>
- [2] Philip Wadler, “The essence of functional programming”
19th Annual Symposium on Principles of Programming Languages (invited talk), 1992 年 1 月
- [3] Philip Wadler, “Monads for functional programming”
Program Design Calculi, Proceedings of the Marktoberdorf Summer School, 1992 年 7-8 月
- [4] Ralf Hinze, “Prological Features in a Functional Setting Axioms and Implementations”
Third Fuji International Symposium on Functional and Logic Programming, 1998 年
- [5] Oleg Kiselyov, Chung-chieh Shan, Daniel P. Friedman and Amr Sabry, “Backtracking, interleaving, and terminating monad transformers (functional pearl)”
ACM SIGPLAN Notices (Vol. 40, No. 9, pp. 192-203), ACM, 2005 年 9 月

